# Maintainer: Kyle Keen <keenerd@gmail.com>

pkgname=jupyterlab_pygments
pkgver=0.2.2
pkgrel=2
pkgdesc="Pygments theme using JupyterLab CSS variables"
arch=('any')
#url="https://pypi.org/project/jupyterlab-pygments/"
url="https://github.com/jupyterlab/jupyterlab_pygments"
license=('BSD')
depends=('python-pygments' 'jupyterlab')
makedepends=('python-build' 'python-installer')
source=(https://pypi.io/packages/source/j/$pkgname/$pkgname-$pkgver.tar.gz)
sha256sums=('7405d7fde60819d905a9fa8ce89e4cd830e318cdad22a0030f7a901da705585d')

build() {
  cd $pkgname-$pkgver
  python -m build --wheel --no-isolation
}

package() {
  cd $pkgname-$pkgver
  python -m installer --destdir="$pkgdir" dist/*.whl

  install -Dm644 "LICENSE" "$pkgdir/usr/share/licenses/$pkgname/LICENSE"
}
